monitor Disk_Interface {
  int position = -2, c = 0, n = 1, args = 0, results = 0;
  cond scan[2];
  cond args_stored, results_stored, results_retrieved;
  argType arg_area; resultType result_area;
  procedure use_disk(int cyl; argType transfer_params;
                     resultType &result_params) {
    if (position == -1)
      position = cyl;
    elseif (position != -1 and cyl > position)
      wait(scan[c],cyl);
    else
      wait(scan[n],cyl);
    arg_area = transfer_params;
    args = args+1; signal(args_stored);
    while (results == 0) wait(results_stored);
    result_params = result_area;
    results = results-1; signal(results_retrieved);
  }
  procedure get_next_request(argType &transfer_params) {
    int temp;
    if (!empty(scan[c]))
      position = minrank(scan[c]);
    elseif (empty(scan[c]) && !empty(scan[n])) {
      temp = c; c = n; n = temp; # swap c and n
      position = minrank(scan[c]);
      }
    else
      position = -1;
    signal(scan[c]);
    while (args == 0) wait(args_stored);
    transfer_params = arg_area; args = args-1;
  }
  procedure finished_transfer(resultType result_vals) {
    result_area := result_vals; results = results+1;
    signal(results_stored);
    while (results > 0) wait(results_retrieved);
  }
}

Figure 5.16 Disk interface monitor.

Copyright © 2000 by Addison Wesley Longman, Inc.